



# Verifpal

*Cryptographic protocol analysis for  
students and engineers*



Nadim Kobeissi  
Symbolic Software, NYU Paris  
*NGI Forum, Helsinki, September 25, 2019*

# What is Formal Verification?

- Using software tools in order to obtain guarantees on the security of cryptographic components.
- Protocols have unintended behaviors when confronted with an active attacker: formal verification can prove security under certain active attacker scenarios!
- Primitives can act in unexpected ways given certain inputs: formal verification: formal verification can prove functional correctness of implementations!

# Formal Verification Today

## Code and Implementations: F\*

- Exports type checks to the Z3 theorem prover.
- Can produce provably functionally correct software implementations of primitives (e.g. Curve25519 in HACL\*).
- Can produce provably functionally correct protocol implementations (Signal\*).

## Protocols: ProVerif, Tamarin

- Take models of protocols (Signal, TLS) and find contradictions to queries.
- “*Can the attacker decrypt Alice’s first message to Bob?*”
- Are limited to the “symbolic model”, CryptoVerif works in the “computational model”.

# Symbolic Verification is Wonderful

- Many papers published in the past 4 years: symbolic verification proving (and finding attacks) in Signal, TLS 1.3, Noise, Scuttlebutt, Bluetooth, 5G and much more!
- This is a great way to work, allowing practitioners to reason better about their protocols before/as they are implemented.


*Why isn't it used more?*

# Tamarin and ProVerif: Examples

```
rule Get_pk:
  [ !Pk(A, pk) ]
  -->
  [ Out(pk) ]

// Protocol
rule Init_1:
  [ Fr(~ekI), !Ltk($I, ltkI) ]
  -->
  [ Init_1( $I, $R, ~ekI )
    , Out( <$I, $R, 'g' ^ ~ekI, sign{'1', $I, $R, 'g' ^ ~ekI
}ltkI> ) ]

rule Init_2:
  let Y = 'g' ^ z // think of this as a group element check
  in
  [ Init_1( $I, $R, ~ekI )
    , !Pk($R, pk(ltkR))
    , In( <$R, $I, Y, sign{'2', $R, $I, Y }ltkR> )
  ]
  --[ SessionKey($I,$R, Y ^ ~ekI)
    , ExpR(z)
  ]->
  [ InitiatorKey($I,$R, Y ^ ~ekI) ]
```



```
letfun writeMessage_a(me:principal, them:principal,
hs:handshakestate, payload:bitstring, sid:sessionid) =
  let (ss:symmetricstate, s:keypair, e:keypair, rs:key,
re:key, psk:key, initiator:bool) =
handshakestateunpack(hs) in
  let (ne:bitstring, ns:bitstring, ciphertext:bitstring)
= (empty, empty, empty) in
  let e = generate_keypair(key_e(me, them, sid)) in
  let ne = key2bit(getpublickey(e)) in
  let ss = mixHash(ss, ne) in
  let ss = mixKey(ss, getpublickey(e)) in
  let ss = mixKey(ss, dh(e, rs)) in
  let s = generate_keypair(key_s(me)) in
```

[...]

```
event(RecvMsg(bob, alice, stagepack_c(sid_b), m)) ==>
(event(SendMsg(alice, c, stagepack_c(sid_a), m))) ||
((event(LeakS(phase0, alice))) && (event(LeakPsk(phase0,
alice, bob)))) || ((event(LeakS(phase0, bob))) &&
(event(LeakPsk(phase0, alice, bob))));
```



# Verifpal: A New Symbolic Verifier

---

1. An intuitive language for modeling protocols (**scientific contribution: a new method for reasoning about protocols in the symbolic model.**)
2. Modeling that avoids user error.
3. Analysis output that's easy to understand.
4. Integration with developer workflow.



# What Are Verifpal's End Goals?

- High quality, robust protocol modeling and analysis for engineers, with integration and live prototyping inside Visual Studio Code.
- High quality educational materials for protocol analysis in undergraduate classes.



# A New Approach to Symbolic Verification

## User-focused approach...

- An intuitive language for modeling protocols.
- Modeling that avoids user error.
- Analysis output that's easy to understand.
- Integration with developer workflow.

## ...without losing strength

- Can reason about advanced protocols (eg. Signal, Noise) out of the box.
- Can analyze for forward secrecy, key compromise impersonation and other advanced queries.
- Unbounded sessions, fresh values, and other cool symbolic model features.



# Verifpal Language

- Explicit principals with discrete internal states (Alice, Bob, Client, Server...)
- Reads like a protocol diagram.
- You don't need to know the language to understand it!
  - *Knows* for private and public values.
  - *Generates* for private fresh values.
  - Assignments.

New Principal: Alice

```
principal Alice[
  knows public c0, c1
  knows private m1
  generates a
]
```

New Principal: Bob

```
principal Bob[
  knows public c0, c1
  knows private m2
  generates b
  gb = G^b
]
```

# Verifpal Language

- Explicit principals with discrete internal states (Alice, Bob, Client, Server...)
- Reads like a protocol diagram.
- You don't need to know the language to understand it!
  - Constants are immutable.
  - Global namespace.
  - Constant cannot reference other constants.

New Principal: Alice

```
principal Alice[  
  knows public c0, c1  
  knows private m1  
  generates a  
]
```

New Principal: Bob

```
principal Bob[  
  knows public c0, c1  
  knows private m2  
  generates b  
  gb = G^b  
]
```


# Verifpal Language: Primitives

- Unlike ProVerif, primitives are *built-in*.
  - Users cannot define their own primitives.
  - Bug, not a feature: eliminate user error on the primitive level.
  - Verifpal not targeting users interested in their own primitives (use ProVerif, it's great!)
- **ENC**(key, plaintext): ciphertext. Symmetric encryption, similar for example to AES-CBC or to ChaCha20.
  - **DEC**(key, **ENC**(key, plaintext)): plaintext. Symmetric decryption.
  - **AEAD\_ENC**(key, plaintext, ad): ciphertext. Authenticated encryption with associated data. ad represents an additional payload that is not encrypted, but that must be provided exactly in the decryption function for authenticated decryption to succeed. Similar for example to AES-GCM or to ChaCha20-Poly1305.
  - **AEAD\_DEC**(key, **AEAD\_ENC**(key, plaintext, ad), ad): plaintext. Authenticated decryption with associated data. See §3.4.4 below for information on how to validate successfully authenticated decryption.

# Verifpal Language: Simple and Intuitive

## Simple Protocol

```
attacker[active]
principal Bob[]
principal Alice[
  generates a
  ga = G^a
]
Alice -> Bob: ga
principal Bob[
  knows private m1
  generates b
  gb = G^b
  e1 = AEAD_ENC(ga^b, m1, gb)
]
Bob -> Alice: gb, e1
principal Alice[
  e1_dec = AEAD_DEC(gb^a, e1, gb)?
]
```



# Passive Attacker

---

- Can observe values as they cross the network.
- Cannot modify values or inject own values.
- Protocol execution happens once.





## Active Attacker

- Can inject own values, substitute values, etc.
- Unbounded protocol executions.
- Keeps learned values between sessions (except if constructed from fresh values.)

# Signal in Verifpal: State Initialization

- Alice wants to initiate a chat with Bob.
- Bob's signed pre-key and one-time pre-key are modeled.

Signal: Initializing Alice and Bob as Principals

```
attacker[active]
principal Alice[
  knows public c0, c1, c2, c3, c4
  knows private alongterm
  galongterm = G^alongterm
]
principal Bob[
  knows public c0, c1, c2, c3, c4
  knows private blongterm, bs
  generates bo
  gblongterm = G^blongterm
  gbs = G^bs
  gbo = G^bo
  gbssig = SIGN(blongterm, gbs)
]
```

# Signal in Verifpal: Key Exchange

- Alice receives Bob's key information and derives the master secret.

## Signal: Alice Initiates Session with Bob

Bob -> Alice: [gblongterm], gbssig, gbs, gbo

```
principal Alice[
  generates ae1
  gae1 = G^ae1
  amaster = HASH(c0, gbs^alongterm, gblongterm^ae1, gbs^ae1, gbo^ae1)
  arkba1, ackba1 = HKDF(amaster, c1, c2)
]
```



# Signal in Verifpal: Messaging

## Signal: Alice Encrypts Message 1 to Bob

```
principal Alice[
  generates m1, ae2
  gae2 = G^ae2
  valid = SIGNVERIF(gblongterm, gbs, gbssig)?
  akshared1 = gbs^ae2
  arkab1, ackab1 = HKDF(akshared1, arkba1, c2)
  akenc1, akenc2 = HKDF(HMAC(ackab1, c3), c1, c4)
  e1 = AEAD_ENC(akenc1, m1, HASH(galongterm, gblongterm, gae2))
]
Alice -> Bob: [galongterm], gae1, gae2, e1
```

## Signal: Bob Decrypts Alice's Message 1

```
principal Bob[
  bkshared1 = gae2^bs
  brkab1, bckab1 = HKDF(bkshared1, brkba1, c2)
  bkenc1, bkenc2 = HKDF(HMAC(bckab1, c3), c1, c4)
  m1_d = AEAD_DEC(bkenc1, e1, HASH(galongterm, gblongterm, gae2))
]
```

# Signal in Verifpal: Queries and Results

- Typical confidential and authentication queries for messages sent between Alice and Bob.
- All queries pass! No contradictions!
- Not surprising: Signal is correctly modeled, long-term public keys are guarded; signature verification is checked.

## Signal: Confidentiality and Authentication Queries

```
queries[
  confidentiality? m1
  authentication? Alice -> Bob: e1
  confidentiality? m2
  authentication? Bob -> Alice: e2
  confidentiality? m3
]
```

## Signal: Initial Analysis Results

```
Verifpal! verification completed at 12:36:53
```

# Protocols Analyzed with Verifpal



- Signal secure messaging protocol.
- Scuttlebutt decentralized protocol.
- ProtonMail encrypted email service.
- Telegram secure messaging protocol.

```
fish /Users/nadim/Documents/git/verifpal

Analysis! HKDF(HMAC(bckba2, c3), c1, c4) now conceivable by reconstructing with HMAC(bckba2, c3), c1, c4
Deduction! m2 found by attacker by deconstructing AEAD_ENC(bkenc3, m2, HASH(gblongterm, galongterm, gbe)) with HKDF(HMAC(bckba2, c3), c1, c4) (depth 5)
Deduction! bkenc3 found by attacker by reconstructing with HMAC(bckba2, c3), c1, c4 (depth 6)
Deduction! brkab1 found by attacker by equivocating with HKDF(bkshared1, brkba1, c2) (depth 13)
Deduction! brkba2 found by attacker by equivocating with HKDF(bkshared2, brkab1, c2) (depth 14)
Deduction! bkshared1 found by attacker by reconstructing with g^attacker_0 (depth 16)
Deduction! bkshared2 found by attacker by reconstructing with g^attacker_0 (depth 17)
Deduction! bkshared1 resolves to gae2^bs (depth 19)
Deduction! galongterm^bs found by attacker by equivocating with bkshared1 (depth 20)
Deduction! gae1^bs found by attacker by equivocating with bkshared1 (depth 20)
Deduction! bkshared2 resolves to gae2^be (depth 21)
Deduction! m2 is obtained by the attacker as m2
Deduction! e2, sent by Attacker and not by Bob and resolving to AEAD_ENC(bkenc3, m2, HASH(gblongterm, galongterm, gbe)), is used in primitive AEAD_DEC(akenc3, e2, HASH(gblongterm, galongterm, gbe)) in Alice's state
  Result! confidentiality? m1: m1 is obtained by the attacker as m1
  Result! authentication? Alice -> Bob: e1: e1, sent by Attacker and not by Alice and resolving to AEAD_ENC(akenc1, m1, HASH(galongterm, gblongterm, gae2)), is used in primitive AEAD_DEC(bkenc1, e1, HASH(galongterm, gblongterm, gae2)) in Bob's state
  Result! confidentiality? m3: m3 is obtained by the attacker as m3
  Result! authentication? Alice -> Bob: e3: e3, sent by Attacker and not by Alice and resolving to AEAD_ENC(akenc5, m3, HASH(gblongterm, galongterm, gae3)), is used in primitive AEAD_DEC(bkenc5, e3, HASH(gblongterm, galongterm, gae3)) in Bob's state
  Result! confidentiality? m2: m2 is obtained by the attacker as m2
  Result! authentication? Bob -> Alice: e2: e2, sent by Attacker and not by Bob and resolving to AEAD_ENC(bkenc3, m2, HASH(gblongterm, galongterm, gbe)), is used in primitive AEAD_DEC(akenc3, e2, HASH(gblongterm, galongterm, gbe)) in Alice's state
Verifpal! verification completed at 21:27:01
REMINDER: Verifpal is experimental software and may miss attacks.
[nadim@nadimsmac:~/D/g/verifpal]-[21:27:01]-[G:master=]
└─$
```

# Verifpal in the Classroom

- Verifpal User Manual: easiest way to learn how to model and analyze protocols on the planet.
- NYU test run: huge success. 20-year-old American undergraduates with **no background whatsoever in security** were modeling protocols in the first two weeks of class and understanding security goals/analysis results.



```
Example Equations
principal Server{
  generates x
  generates y
  gx = G^x
  gy = G^y
  gxy = gx*y
  gyx = gy*x
}
```

In the above,  $gxy$  and  $gyx$  are considered equivalent by Verifpal. In Verifpal, all equations must have the constant  $G$  as their root generator. This mirrors Diffie-Hellman behavior. Furthermore, all equations can only have two constants ( $a^*b$ ), but as we can see above, equations can be built on top of other equations (as in the case of  $gxy$  and  $gyx$ ).


## 2.6 MESSAGES

Sending messages over the network is simple. Only constants may be sent within messages:

```
Example: Messages
Alice -> Bob: ga, e1
Bob -> Alice: [gb], e2
```

Let's look at the two messages above. In the first, Alice is the sender and Bob is the recipient. Notice how Alice is sending Bob her long-term public key  $ga = G^a$ . An active attacker could intercept  $ga$  and replace it with a value that they control. But what if we want to model our protocol such that Alice has pre-authenticated<sup>2</sup> Bob's public key  $gb = G^b$ ? This is where *guarded constants* become useful.

<sup>2</sup>"Pre-authentication" refers to Alice confirming the value of Bob's public key before the protocol session begins. This helps avoid having an active attacker trick Alice to use a fake public key for Bob. This fake public key could instead be the attacker's own public key.






# Verifpal in the Classroom

- Upcoming Eurocrypt 2020 affiliated event:  
<https://verifpal.com/eurocrypt2020/> – Verifpal tutorial!
- Verifpal has a place in your undergraduate classroom and will do a better job teaching students about protocols and models than anything else in the world.



International Association  
for Cryptologic Research





# Verifpal's Role in the NGI Vision

- Provide engineers, developers and students with the accessibility they need for the analysis of critical cryptographic systems and designs.
- Broaden access to the latest research into better understanding the security of cryptographic systems in software.



# Try Verifpal Today

*Verifpal is released as free and open source software, under version 3 of the GPL.*

Check out Verifpal today:

[verifpal.com](https://verifpal.com)

Support Verifpal development:

[verifpal.com/donate](https://verifpal.com/donate)

